Choice of the parameters

Based on the computations of N. Bacaër in: Un modèle mathématique des débuts de l'épidémie de coronavirus en France
The starting dates are T = March 15, 2020 and initial values are labelled ST, ET, etc.
Parameters of the SEIR model are denoted by a, b, c as in the above paper and correspond to β, a and γ with standard notations

In[27]:=
```
NTot = 67.8 * 10^6;
a = 2.33;
b = 0.25;
c = 1;
R0 = a/c;
ET = 5970 / NTot;
IT = 1278 / NTot;
RT = (ET + IT)/(R0 - 1);
ST = 1 - R0 RT;
```

SEIR model with a *q* parameter

In[36]:=
```
F[q_, T_, PS_] :=
 Module[{M = {Ex[s], Inf[s], R[s]} /. NDSolve[{S'[t] == -a/q Inf[t] S[t], Ex'[t] ==
      a/q Inf[t] S[t] - b Ex[t], Inf'[t] == b Ex[t] - c Inf[t], R'[t] == c Inf[t],
     S[0] == ST, Ex[0] == ET, Inf[0] == IT, R[0] == RT}, {S, Ex, Inf, R}, {t, 0, T}]},
  Plot[M, {s, 0, T}, PlotStyle -> {{Blue, PS}, {Red, PS}, {Darker[Green], PS}}]]
```

Without confinement

In[39]:=
```
Show[F[1, 100, Thick],
 ListLinePlot[{{60, 0.15}, {70, 0.15}}, PlotStyle -> {Darker[Green], Thick}],
 ListLinePlot[{{60, 0.125}, {70, 0.125}}, PlotStyle -> {Blue, Thick}],
 ListLinePlot[{{60, 0.1}, {70, 0.1}}, PlotStyle -> {Red, Thick}]
 , PlotRange -> {All, {0, 0.18}}]
```

Out[39]=



With confinement

In[41]:=
```
Show[F[1.7, 200, Thick],
 ListLinePlot[{{110, 0.15}, {130, 0.15}}, PlotStyle → {Darker[Green], Thick}],
 ListLinePlot[{{110, 0.125}, {130, 0.125}}, PlotStyle → {Blue, Thick}],
 ListLinePlot[{{110, 0.1}, {130, 0.1}}, PlotStyle → {Red, Thick}]
 , PlotRange → {All, {0, 0.18}}]
```

Out[41]=



The percentage of R(Tstop) at Tstop=60 as a function of q

```
Off[NDSolve::ndnum]
Off[ReplaceAll::reps]
```

```
Fasymp[Tstop_] := Module[
  {M = R[Tstop] - RT /. NDSolve[{S'[t] == -RR c Inf[t] S[t], Ex'[t] == RR c Inf[t] S[t] -
        b Ex[t], Inf'[t] == b Ex[t] - c Inf[t], R'[t] == c Inf[t], S[0] == ST,
      Ex[0] == ET, Inf[0] == IT, R[0] == RT}, {S, Ex, Inf, R}, {t, 0, Tstop}]},
  Plot[M, {RR, 0, 5}, PlotRange → All, PlotStyle → Darker[Green]]]
```

```
Show[Fasymp[30], Fasymp[60], Fasymp[90], Plot[1 + ProductLog[-e^(-RR) RR]/RR - RT,
  {RR, 0, 5}, PlotPoints → 40, PlotStyle → {Darker[Green], Thick}, PlotRange → All]]
```



SEIR model with two populations with different parameters *q*

In[42]:=
```
FEIRmodif[p_, q1_, q2_, T_] :=
 Show[Module[{M = {Ex1[s] + Ex2[s], Inf1[s] + Inf2[s], R1[s] + R2[s], R2[s]} /.
     NDSolve[{S1'[t] == -a (Inf1[t] + Inf2[t])/q1 S1[t], S2'[t] ==
         -a (Inf1[t] + Inf2[t])/q2 S2[t], Ex1'[t] == a (Inf1[t] + Inf2[t])/q1 S1[t] - b Ex1[t],
        Ex2'[t] == a (Inf1[t] + Inf2[t])/q2 S2[t] - b Ex2[t], Inf1'[t] == b Ex1[t] - c Inf1[t],
        Inf2'[t] == b Ex2[t] - c Inf2[t], R1'[t] == c Inf1[t], R2'[t] == c Inf2[t],
        S1[0] == (1 - p) ST, S2[0] == p ST, Ex1[0] == (1 - p) ET, Ex2[0] == p ET,
        Inf1[0] == (1 - p) IT, Inf2[0] == p IT, R1[0] == (1 - p) RT, R2[0] == p RT},
       {S1, S2, Ex1, Ex2, Inf1, Inf2, R1, R2}, {t, 0, T}]},
    Plot[M, {s, 0, T}, PlotStyle → {{Thick, Blue}, {Thick, Red}, {Thick, Darker[Green]},
       {Thick, Dashed, Darker[Green]}}, PlotRange → All]],
  ListLinePlot[{{0, p}, {T, p}}, PlotStyle → {Thick, Dotted, Darker[Green]}]]
```

Fitting the data by the end of March: the curve of N. Bacaër vs a model with two groups

In[43]:=
```
FEIRmodifCases[p_, q1_, q2_, T_, PS_] :=
 Module[{M = {Ex1[s], Ex2[s], Inf1[s], Inf2[s], R1[s], R2[s]} /.
     NDSolve[{S1'[t] == -a (Inf1[t] + Inf2[t])/q1 S1[t], S2'[t] == -a (Inf1[t] + Inf2[t])/q2 S2[t],
        Ex1'[t] == a (Inf1[t] + Inf2[t])/q1 S1[t] - b Ex1[t],
        Ex2'[t] == a (Inf1[t] + Inf2[t])/q2 S2[t] - b Ex2[t], Inf1'[t] == b Ex1[t] - c Inf1[t],
        Inf2'[t] == b Ex2[t] - c Inf2[t], R1'[t] == c Inf1[t], R2'[t] == c Inf2[t],
        S1[0] == (1 - p) ST, S2[0] == p ST, Ex1[0] == (1 - p) ET, Ex2[0] == p ET,
        Inf1[0] == (1 - p) IT, Inf2[0] == p IT, R1[0] == (1 - p) RT, R2[0] == p RT},
       {S1, S2, Ex1, Ex2, Inf1, Inf2, R1, R2}, {t, 0, T}]},
  Plot[Log[M[[1]][[3]] + M[[1]][[4]] + M[[1]][[5]] + M[[1]][[6]]],
   {s, 0, T}, PlotStyle → PS, PlotRange → All]]
```

In[46]:=
```
FitFEIR[p_, q_, q1_, q2_, T_] :=
 Show[FEIRmodifCases[p, q1, q2, T, {Brown, Thickness[0.025], Opacity[0.6]}],
  FEIRmodifCases[p, q, q, T, {Black, Thick, Dotted}]]
```

In[51]:=
```
q /. Solve[0 == (1 - p)/2.35 + p/q - 1/1.7 /. p → 0.02, q][[1]]
Show[FitFEIR[0.02, 1.7, 2.35, %, 15],
 ListLinePlot[{{9, -8.25}, {11, -8.25}}, PlotStyle → {Black, Thick, Dotted}],
 ListLinePlot[{{9, -8.5}, {11, -8.5}},
  PlotStyle → {Brown, Thickness[0.025], Opacity[0.6]}]]
```

Out[51]= 0.116813

Out[52]=



In[65]:=
```
Show[F[1.7, 200, Thick], PlotRange → {All, {0, 0.032}}]
Show[FEIRmodif[0.02, 2.35, 0.117, 200], ListLinePlot[{{100, 0.0125}, {128, 0.0125}},
  PlotStyle → {Darker[Green], Thick, Dashed}], PlotRange → {All, {0, 0.032}}]
```

Out[65]=



Out[66]=



Phase transitions

```
Off[Power::infy]
```

```
f[p_, q1_, q2_] := r /. FindRoot[(1 - p) ST ⅇ^((R0 (RT-r))/q1) + p ST ⅇ^((R0 (RT-r))/q2) + r == 1, {r, 0.5}]

Rq[p_, q1_, q2_] := R0 ((1 - p)/q1 + p/q2)

Fasymp2[p_, q1_, q2_, Tstop_] :=
 R1[Tstop] + R2[Tstop] /. NDSolve[{S1'[t] == -a (Inf1[t] + Inf2[t])/q1 S1[t],
     S2'[t] == -a (Inf1[t] + Inf2[t])/q2 S2[t], Ex1'[t] == a (Inf1[t] + Inf2[t])/q1 S1[t] - b Ex1[t],
     Ex2'[t] == a (Inf1[t] + Inf2[t])/q2 S2[t] - b Ex2[t], Inf1'[t] == b Ex1[t] - c Inf1[t],
     Inf2'[t] == b Ex2[t] - c Inf2[t], R1'[t] == c Inf1[t], R2'[t] == c Inf2[t],
     S1[0] == (1 - p) ST, S2[0] == p ST, Ex1[0] == (1 - p) ET, Ex2[0] == p ET,
     Inf1[0] == (1 - p) IT, Inf2[0] == p IT, R1[0] == (1 - p) RT, R2[0] == p RT},
   {S1, S2, Ex1, Ex2, Inf1, Inf2, R1, R2}, {t, 0, Tstop}]

P2[Tstop_] := ParametricPlot[{Rq[0.02, 2.35, q], Fasymp2[0.02, 2.35, q, Tstop]},
  {q, 0, 5}, PlotStyle → {Darker[Green]},
  PlotRange → {{0, 5}, {0, 0.5}}, AspectRatio → 1 / GoldenRatio]

Pp[p_, q1_, Tstop_] := ParametricPlot[
  {Rq[p, q1, q], Fasymp2[p, q1, q, Tstop]}, {q, 0, 5}, PlotStyle → {Darker[Green]},
  PlotRange → {{0, 5}, {0, 0.5}}, AspectRatio → 1 / GoldenRatio]

P0 = Plot[1 + ProductLog[-ⅇ^-RR RR]/RR, {RR, 0, 5}, PlotPoints → 40,
   PlotStyle → {Darker[Green], Dotted, Thick}, PlotRange → All];
P1 = ParametricPlot[{Rq[0.02, 2.35, q], f[0.02, 2.35, q]}, {q, 0, 5},
   PlotStyle → {Darker[Green], Thick}, PlotRange → {{0, 5}, {0, 0.5}}];

Show[P0, P1, P2[30], P2[60], P2[90],
 AspectRatio → 1 / GoldenRatio, PlotRange → {{0, 5}, {0, 1}}]
```